Module 3.3: nag_err_fun

Error Functions

`nag_err_fun` contains procedures for approximating the error function, the complementary error function and Dawson's integral.

Contents

Introduction

This module contains procedures for approximating the error function erf x and related functions.

- `nag_erf` computes the error function

$$\operatorname{erf} x = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt.$$

- `nag_erfc` computes the complementary error function

$$\operatorname{erfc} x = \frac{2}{\sqrt{\pi}} \int_x^\infty e^{-t^2} dt \ = \ 1 - \operatorname{erf} x.$$

- `nag_dawson` computes Dawson's integral

$$F(x) = e^{-x^2} \int_0^x e^{t^2} dt.$$

Further details of these special functions may be found in Abramowitz and Stegun [1].

In general the approximations are based on expansions in terms of Chebyshev polynomials $T_r(t) = \cos(r \arccos t)$, where $t = t(x)$ is a mapping from the region of interest to the interval $[-1, 1]$, on which the Chebyshev polynomials are defined. Further details appear in Section 6.1 of the individual procedure documents.

Procedure: nag_erf

1 Description

`nag_erf` evaluates the error function

$$\operatorname{erf} x = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2}\,dt.$$

2 Usage

```
USE nag_err_fun
[value =] nag_erf(x)
```

The function result is a scalar, of type real(kind=*wp*), containing $\operatorname{erf} x$.

3 Arguments

3.1 Mandatory Argument

x — real(kind=*wp*), intent(in)
Input: the argument x of the function.

4 Error Codes

None.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Algorithmic Detail

- For $|x| \leq 2$, the procedure uses a Chebyshev expansion of the form

  $$\operatorname{erf} x = x \sum_{r=0}{}' a_r T_r(t),$$

  where $t = x^2/2 - 1$.

- For $2 < |x| < x_{\mathrm{hi}}$,

  $$\operatorname{erf} x = \operatorname{sign} x \left(1 - \frac{e^{-x^2}}{|x|\sqrt{\pi}} \sum_{r=0}{}' b_r T_r(t)\right),$$

  where $t = (x-7)/(x+3)$.

- For $|x| \geq x_{\mathrm{hi}}$,

  $$\operatorname{erf} x = \operatorname{sign} x.$$

The parameter x_{hi} is the value above which $\operatorname{erf} x = \pm 1$ to within `EPSILON(1.0_wp)`. Its value is given in the Users' Note for your implementation.

6.2 Accuracy

On a machine with approximately 11 significant figures the procedure agrees with available tables to 10 figures and consistency checking with the procedure nag_erfc of the relation

$$\operatorname{erf} x + \operatorname{erfc} x = 1.0$$

shows errors in at worst the 11th figure.

Procedure: nag_erfc

1 Description

nag_erfc calculates an approximate value for the complementary error function

$$\operatorname{erfc} x = \frac{2}{\sqrt{\pi}} \int_x^\infty e^{-t^2} dt \;=\; 1 - \operatorname{erf} x.$$

2 Usage

```
USE nag_err_fun
[value =] nag_erfc(x)
```

The function result is a scalar, of type real(kind=wp), containing erfc x.

3 Arguments

3.1 Mandatory Argument

x — real(kind=wp), intent(in)

Input: the argument x of the function.

4 Error Codes

None.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Algorithmic Detail

- For $x \geq 0$, the procedure uses a Chebyshev expansion of the form

  $$\operatorname{erfc} x = e^{-x^2} y(x),$$

  where $y(x) = \sum_{r=0}{}' a_r T_r(t)$ and $t = (x - 3.75)/(x + 3.75)$, $-1 \leq t \leq +1$.
- For $x \geq x_{\rm hi}$, where there is a danger of setting underflow, the result is returned as zero.
- For $x < 0$, erfc $x = 2 - e^{-x^2} y(|x|)$.
- For $x < x_{\rm low} < 0$, the result is returned as 2.0 which is correct to within EPSILON(1.0_wp).

The values of $x_{\rm hi}$ and $x_{\rm low}$ are given in the Users' Note for your implementation.

6.2 Accuracy

If δ and ε are relative errors in the argument and result respectively, then in principle

$$|\varepsilon| \simeq |\theta\delta|, \quad \text{where} \quad \theta = \frac{2xe^{-x^2}}{\sqrt{\pi}\,\mathrm{erfc}\; x}.$$

That is, the relative error in the argument x is amplified by a factor θ in the result. The behaviour of this factor is shown in Figure 1.



Figure 1: The relative error amplification factor θ.

It should be noted that near $x = 0$ this factor behaves as $2x/\sqrt{\pi}$ and hence the accuracy is largely determined by `EPSILON(1.0_wp)`. Also for large negative x, where the factor is $\sim xe^{-x^2}/\sqrt{\pi}$, accuracy is mainly limited by `EPSILON(1.0_wp)`. However, for large positive x, the factor becomes $\sim 2x^2$ and to an extent relative accuracy is necessarily lost. The absolute accuracy E is given by

$$E \simeq \frac{2xe^{-x^2}}{\sqrt{\pi}}\delta,$$

so absolute accuracy is guaranteed for all x.

Procedure: nag_dawson

1 Description

nag_dawson evaluates an approximation for Dawson's integral

$$F(x) = e^{-x^2} \int_0^x e^{t^2} dt.$$

2 Usage

```
USE nag_err_fun
[value =] nag_dawson(x)
```

The function result is a scalar, of type real(kind=*wp*), containing $F(x)$.

3 Arguments

3.1 Mandatory Argument

x — real(kind=*wp*), intent(in)
Input: the argument x of the function.

4 Error Codes

None.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Algorithmic Detail

- For $0 < |x| \leq 4$, the procedure uses a Chebyshev expansion of the form

$$F(x) = x \sum_{r=0}{}' a_r T_r(t), \quad \text{where} \quad t = 2\left(\frac{x}{4}\right)^2 - 1.$$

- For $|x| > 4$,

$$F(x) = \frac{1}{x} \sum_{r=0}{}' b_r T_r(t), \quad \text{where} \quad t = 2\left(\frac{4}{x}\right)^2 - 1.$$

- For $|x|$ near zero, $F(x) \simeq x$, and for $|x|$ large, $F(x) \simeq 1/(2x)$. These approximations are used for those values of x for which the result is correct to EPSILON(1.0_*wp*). For very large x on some machines, $F(x)$ may underflow and then the result is set exactly to zero.

6.2 Accuracy

Let δ and ε be the relative errors in the argument and result respectively.

If δ is considerably greater than EPSILON(1.0_*wp*) (i.e., if δ is due to data errors etc.), then ε and δ are approximately related by:

$$\varepsilon \simeq |\theta|\delta, \quad \text{where} \quad \theta = \frac{x(1-2xF(x))}{F(x)}.$$

The behaviour of the error amplification factor $|\theta|$ is shown in Figure 2.



Figure 2: The relative error amplification factor $|\theta|$.

However, if δ is of the same order as EPSILON(1.0_*wp*), then rounding errors could make ε somewhat larger than the above relation indicates. In fact ε will be largely independent of x or δ, but will be of the order of a few times EPSILON(1.0_*wp*).

Example 1: Evaluation of Error Functions and Dawson's Integral

This example program evaluates the functions `nag_erf`, `nag_erfc` and `nag_dawson` for various values of the argument `x`.

1 Program Text

Note. The listing of the example program presented below is double precision. Single precision users are referred to Section 5.2 of the Essential Introduction for further information.

```
PROGRAM nag_err_fun_ex01

  ! Example Program Text for nag_err_fun
  ! NAG fl90, Release 3. NAG Copyright 1997.

  ! .. Use Statements ..
  USE nag_examples_io, ONLY : nag_std_out
  USE nag_err_fun, ONLY : nag_erf, nag_erfc, nag_dawson
  ! .. Implicit None Statement ..
  IMPLICIT NONE
  ! .. Intrinsic Functions ..
  INTRINSIC KIND
  ! .. Parameters ..
  INTEGER, PARAMETER :: n = 11
  INTEGER, PARAMETER :: wp = KIND(1.0D0)
  ! .. Local Scalars ..
  INTEGER :: i
  REAL (wp) :: dawson_x, erfc_x, erf_x
  ! .. Local Arrays ..
  REAL (wp) :: x(n)
  ! .. Executable Statements ..

  WRITE (nag_std_out,*) 'Example Program Results for nag_err_fun_ex01'

  WRITE (nag_std_out,*)
  WRITE (nag_std_out,*) '      x         erf(x)      erfc(x)      F(x)'
  x = (/ -10.0_wp, -6.0_wp, -4.5_wp, -1.0_wp, -0.5_wp, 0.0_wp, 1.0_wp, &
   2.5_wp, 4.0_wp, 6.0_wp, 10.0_wp/)

  DO i = 1, n

    erf_x = nag_erf(x(i))
    erfc_x = nag_erfc(x(i))
    dawson_x = nag_dawson(x(i))

    WRITE (nag_std_out,fmt='(1X,1P,4E12.3)') x(i), erf_x, erfc_x, dawson_x
  END DO

END PROGRAM nag_err_fun_ex01
```

2 Program Data

None.

3 Program Results

```
Example Program Results for nag_err_fun_ex01

      x          erf(x)       erfc(x)       F(x)
  -1.000E+01  -1.000E+00   2.000E+00  -5.025E-02
  -6.000E+00  -1.000E+00   2.000E+00  -8.454E-02
  -4.500E+00  -1.000E+00   2.000E+00  -1.141E-01
  -1.000E+00  -8.427E-01   1.843E+00  -5.381E-01
  -5.000E-01  -5.205E-01   1.520E+00  -4.244E-01
   0.000E+00   0.000E+00   1.000E+00   0.000E+00
   1.000E+00   8.427E-01   1.573E-01   5.381E-01
   2.500E+00   9.996E-01   4.070E-04   2.231E-01
   4.000E+00   1.000E+00   1.542E-08   1.293E-01
   6.000E+00   1.000E+00   2.152E-17   8.454E-02
   1.000E+01   1.000E+00   2.088E-45   5.025E-02
```

Additional Examples

Not all example programs supplied with NAG *fl*90 appear in full in this module document. The following additional examples, associated with this module, are available.

`nag_err_fun_ex02`

Evaluation of the error function.

`nag_err_fun_ex03`

Evaluation of the complementary error function.

`nag_err_fun_ex04`

Evaluation of Dawson's integral.

References

[1] Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* Dover Publications (3rd Edition)